```
*=============CRYSTAL_PI_Network_AC============================
```

When crystals are used in an adjustable frequency mode, the PI network formed by the crystal and external capacitors C1 and C2 are expected to have 180 degrees phase shift.

Spice seems to like to do an automatic wrap around of phase in its output plots. It is often time much easier to see things without this wrap around.

```
*=========An_Adjustable_Crystal_Oscillator========
OneMegCrystalReal
*
*                  |OUT|
*          |\      |___|
*   _______| \/\____|___/\  /\  /\_
*  |       | /\/            \/  \/  |
*  |       |/          R2 13.7k     |
*  |                                |
*  |        1Meg                    |
*  |___/\  /\  /\___________________|
*        \/  \/     |
*         R1        |
*          ____     |
*  _|_  ||    ||  _|_
* |VA|__||    ||__|VB|
* |___|  ||____||  |___|
*  _|_             _|_   18p
*
*  _|_ C1  16.5p   _|_ C2
*  ///             ///
==================================================================
```

Only one inverter is in the loop. The values of R2,C2, and C2 will be providing a 180 degree phase shift at some frequency. The test circuit is shown below.

```
*=========AC_of_the_PI_Network==================================================
OneMegCrystalPI
*
*        1Meg
*   ___/\  /\  /\______/\  /\  /\__|VC|
*  |     \/  \/     |    \/  \/    |___|
*  |      R1        |      R2       _|_
*  |               |               / _ \
*  |       ____     |             // \ \
*  _|_  ||    ||  _|_             \   \//
* |VA|__||    ||__|VB|             \___/
* |___|  ||____||  |___|  18p        _|_
*  _|_             _|_               ///
*
*  _|_ C1 16.5p    _|_ C2
*  ///             ///
*
*        1Meg
*   ___/\  /\  /\____________________________/\  /\  /\__|VC|
*  |     \/  \/                       |        \/  \/    |___|
*  |      R1                          |         R2        _|_
*  |                                  |       13.7K      / _ \
*  _|_       340      VCS   VLS/\/\/\    _|_           // \ \
* |VA|____/\  /\  /\__||____| () () |__|VB|           \   \//
* |___|  |  \/  \/      ||            |  |___|           \___/
*  _|_17p | RS           CS      LS 3.5H |  _|_  18p       _|_
*         |        3p       7f          |                  ///
*  _|_ C1 |_____||_______________________| _|_ C2
*  ///          ||  CP                    ///
*
* 1.017e+6
* Freq                  Mode  L1     C1        R1    C0      Q
* 1 MHz                 fund  3.5H   0.007pf   340   3pf     64679
*
.OPTIONS  GMIN=1e-18    METHOD=trap   ABSTOL=1e-18  TEMP=27 srcsteps = 1  gminsteps = 1
VIN       VC     0      DC       AC     1
XCRY1     VA     VB     XCRYST
R1        VA     VC     1000k
R2        VC     VB     12.7k
C1        VA     0      16.5p
C2        VB     0      18p
===============================================================
```

At one frequency, the impedance of the crystal will try to look like a short circuit. At another frequency it will try

to look like an open circuit.




```
*=========Run_AC=============================================
.control
*AC        DECLin NUMDEC FSTART      FSTOP
ac         dec    600000 1.0165meg   1.0184meg
set        pensize = 2
plot       mag(va) mag(vb)    ylog
================================================================================
```

But look what happens when one tries to do a phase plot of the AC transfer function. In some cases it might be convenient to wrap the phase around 180 degrees. But what if one does not want to do that?




```
================================================================================
```

Processing the output waveform arrays can handle that.

```
*=========Process_Output_Arrays========
let        lenV= length(va)
compose    phva_deg start = 1 stop = $&lenV step =1
compose    f_Hz     start = 1 stop = $&lenV step =1
settype    frequency       f_Hz

*=========Remove_Phase_Wrap_around========
let        i = 0
let        phat180=180
let        f_at180=1
repeat     $&lenV
let        phva_deg[i] = ph(va[i])
if         (phva_deg[i] > 0)
let        phva_deg[i] = phva_deg[i] -360
endif
if         (abs(phva_deg[i]+180) < phat180)
let        phat180 = abs(phva_deg[i]+180)
```

```
let       f_at180 = frequency[i]
endif
let       f_Hz[i] = frequency[i]
let       i = i +1
end
```

==

Not having the wrap around at times might make things easier on the eyes. Notice that the dB scale has a gain of 5, and that a -180 reference line has been added.




```
*=========Find_Where_Phase_180========
let       phref = -180
plot      phva_deg  phref 5*db(va)  vs f_Hz
plot      phva_deg  phref 5*db(va)  phvb_deg  5*db(vb) vs f_Hz
plot      phva_deg  phref 5*db(va)  phvb_deg  5*db(vb) vs f_Hz xlimit 1.0168Meg 1.0172Meg
plot      phva_deg  phref 5*db(va)  phvb_deg  5*db(vb) vs f_Hz xlimit 1.0170Meg 1.0172Meg ylimit -240 -160
print     f_at18
.endc
```

==

This -180 degree phase shift is resulting from all the elements in the PI network acting together.




```
*           1Meg
*      ___/\  /\  /\______________________________/\  /\  /\__|VC |
*     |      \/  \/                          |        \/  \/    |___|
*     |      R1                              |        R2         _|_
*     |                                      |       13.7K      / _ \
*    _|_        340      VCS   VLS/‾\/‾\/‾\  _|_               //  \ \
*   |VA |____/\  /\  /\__||____| () ()  |__|VB |               \   \//
*   |___|      \/  \/    ||             |  |___|                \___/
*    _|_17p   RS         CS     LS 3.5H |   _|_  18p             _|_
*    ___            3p      7f          |   ___                  ///
*    _|_ C1  |______||__________________|   _|_ C2
*    ///            ||CP                    ///

* 1.017e+6               Z0= 22.3Meg           CP=53K
* Freq                   Mode  L1     CS       R1    CP      Q
```

```
* 1 MHz                fund  3.5H   0.007pf  340   3pf     64679
*
plot      phva_deg  phref  5*db(va) vs f_Hz xlimit 1.0175Meg 1.0180Meg ylimit -240 -160
print     f_at180
.endc
```

==

It is looking like the -180 phase shift is happening close to where the phase at VB is close to zero. So there is about 180 degrees between VB and VA. In other words, the crystal is effectively just an inductor connecting VB to VA.



==

Printing out the 180 degree phase crossing reveals the predicted oscillation frequency.

Circuit: OneMegCrystal_PI_network
f_at180 =   1.01712e+06

And the measured free running oscillation frequency was.

Circuit: OneMegCrystalReal_16.5pf
mean(vref) =   2.42861e+00
mean(1/dt_tp) =   1.01704e+06   0.008% (80Hz)

===========Full_Netlist_For_Copy_Paste======================

```
OneMegCrystalPI
*
*          1Meg
*     ___/\  /\  /\______/\  /\  /\__|VC |
*    |      \/  \/     |     \/  \/   |___|
*    |       R1        |       R2      _|_
*    |                 |              / _ \
*    |         ____    |             // \ \
*   _|_      ||    ||  _|_           \   \//
*  |VA |__ ||    ||__|VB |  18p      \___/
*  |___|   ||____||  |___|            _|_
*   _|_                _|_            ///
*   _|_  C1  16.5p     _|_  C2
*   ///                ///
*
*
*          1Meg
*     ___/\  /\  /\__________________________/\  /\  /\__|VC |
*    |      \/  \/                      |         \/  \/   |___|
*    |       R1                         |          R2       _|_
*    |                                  |         12.7K    / _ \
*   _|_        340      VCS   VLS/‾\/‾\/‾\                 // \ \
*  |VA |____/\  /\  /\__||____| () () |__|VB |             \   \//
*  |___|_17p |  \/  \/   CS     LS 3.5H  |  |___|  18p      \___/
*   _|_      RS     3p   7f              |   _|_            _|_
*   _|_ C1   |_____||____________________|   _|_ C2         ///
```

```
*   ///         || CP                      ///
*
* 1.017e+6
* Freq                 Mode  L1     C1       R1    C0      Q
* 1 MHz                fund  3.5H   0.007pf  340   3pf     64679
*
.OPTIONS  GMIN=1e-18    METHOD=trap   ABSTOL=1e-18  TEMP=27 srcsteps = 1  gminsteps = 1
VIN       VC     0      DC      AC     1
XCRY1     VA     VB     XCRYST
R1        VA     VC     1000k
R2        VC     VB     12.7k
C1        VA     0      16.5p
C2        VB     0      18p

.control
*AC       DECLin NUMDEC FSTART     FSTOP
ac        dec    600000 1.0165meg  1.0184meg
set       pensize = 2
plot      mag(va) mag(vb)    ylog

let       lenV= length(va)
compose   phva_deg start = 1 stop = $&lenV step =1
compose   phvb_deg start = 1 stop = $&lenV step =1
compose   f_Hz     start = 1 stop = $&lenV step =1
settype   frequency       f_Hz
*=========Remove_Phase_Wrap_around========
let       i = 0
let       phat180=180
let       f_at180=1

repeat    $&lenV
let       phva_deg[i] = ph(va[i])
let       phvb_deg[i] = ph(vb[i])


if        (phva_deg[i] > 0)
let       phva_deg[i] = phva_deg[i] -360
endif

if        (phvb_deg[i] > 0)
let       phvb_deg[i] = phvb_deg[i] +360-360
endif

if        (abs(phva_deg[i]+180) < phat180)
let       phat180 = abs(phva_deg[i]+180)
let       f_at180 = frequency[i]
endif

let       f_Hz[i] = frequency[i]
let       i = i +1
end

*=========Find_Where_Phase_180========

let       phref = -180
plot      phva_deg  phvb_deg phref  5*db(va) 5*db(vb) vs f_Hz
plot      phva_deg  phvb_deg phref  5*db(va) 5*db(vb) vs f_Hz xlimit 1.0168Meg 1.0172Meg
plot      phva_deg  phref  5*db(va) vs f_Hz xlimit 1.0175Meg 1.0180Meg ylimit -240 -160
print     f_at180
.endc

*=========CRYSTAL_SubCircuit============================================
.SUBCKT   XCRYST VA     VB
 RS       VA     VCS    340
 CS       VCS    VLS    7f
 LS       VLS    VB     3.5
 CP       VA     VB     3pf
.ENDS     XCRYST
*                             _  _  _
*    ___             VCS  VLS/ \/ \/ \    ___
*   |VA |__/\  /\  /\__||____| () () |___|VB |
*   |___|  |\/  \/     ||               |  |___|
*          |  RS        CS     LS       |   _|_
*          |                            |   ///
*          |____||______________________|
*               || CP
*                                    CS LS = 29.97meg @ 885Gohms
*
*                      ___
*              ___   ||   ||  ___
*             |VA |__||   ||__|VB |
*             |___|  ||   ||  |___|
*                    ||___||
*
* Freq                 Mode  L1     C1       R1    C0      Q
```

```
* 1 MHz                fund  3.5H   0.007pf  340   3pf     64679
*
*=====================================================================
.end
```

4.11.10_4.54PM
dsauersanjose@aol.com
Don Sauer
http://www.idea2ic.com/